



Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

**A MODULAR THREE-DIMENSIONAL
FINITE-DIFFERENCE GROUND-WATER
FLOW MODEL**

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological
Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

Narrative for Module RIV1FM


This module adds terms representing river leakage to the accumulators HCOF and RHS.

1. If NRIVER is less than or equal to zero, in the current stress period, there are no river reaches. RETURN.
2. For each reach in the RIVR list, DO STEPS 3-8.
3. Determine the column (IC), row (IR), and layer (IL).
4. If the cell is external ($IBOUND(IC, IR, IL) \leq 0$), bypass processing on this reach and go on to the next reach.
5. Since the cell is internal, get the river data (riverhead conductance of the riverbed and elevation of the bottom of the riverbed).
6. Compare the head in the aquifer (HNEW) to the elevation of the bottom of the riverbed (RBOT).
7. If the head in the aquifer (HNEW) is greater than the elevation of the bottom of the riverbed (RBOT), add the term $-CRIV*HRIV$ to the accumulator RHS and the term $-CRIV$ to the accumulator HCOF. (CRIV is the riverbed conductance; HRIV is the riverhead.)
8. If the head in the aquifer (HNEW) is less than or equal to RBOT, add the term $-CRIV*(HRIV - RBOT)$ to the accumulator RHS.
9. RETURN.

Flow Chart for Module RIV1FM

RHS is an accumulator in which the right hand side of the equation is formulated.

HCOF is an accumulator in which the coefficient of head in the cell is formulated.



```

SUBROUTINE RIV1FM(NRIVER, MXRIVR, RIVR, HNEW, HCOF, RHS, IBOUND,
1          NCOL, NROW, NLAY)
C
C-----VERSION 0915 27AUG1982 RIV1FM
C *****:*****
C ADD RIVER TERMS TO RHS AND HCOF
C *****:*****
C
C SPECIFICATIONS:
C -----
C
C DOUBLE PRECISION HNEW
C DIMENSION RIVR(6, MXRIVR), HNEW(NCOL, NROW, NLAY),
1          HCOF(NCOL, NROW, NLAY), RHS(NCOL, NROW, NLAY),
2          IBOUND(NCOL, NROW, NLAY)
C -----
C
C
C1-----IF NRIVER<=0 THERE ARE NO RIVERS. RETURN.
          IF(NRIVER.LE.0)RETURN
C
C2-----PROCESS EACH CELL IN THE RIVER LIST.
          DO 100 L=1, NRIVER
C
C3-----GET COLUMN, ROW, AND LAYER OF CELL CONTAINING REACH
          IL=RIVR(1, L)
          IR=RIVR(2, L)
          IC=RIVR(3, L)
C
C4-----IF THE CELL IS EXTERNAL SKIP IT.
          IF(IBOUND(IC, IR, IL).LE.0)GO TO 100
C
C5-----SINCE THE CELL IS INTERNAL GET THE RIVER DATA.
          HRIV=RIVR(4, L)
          CRIV=RIVR(5, L)
          RBOT=RIVR(6, L)
          HHNEW=HNEW(IC, IR, IL)
C
C6-----COMPARE AQUIFER HEAD TO BOTTOM OF STREAM BED.
          IF(HHNEW.LE.RBOT)GO TO 96
C
C7-----SINCE HEAD>BOTTOM ADD TERMS TO RHS AND HCOF.
          RHS(IC, IR, IL)=RHS(IC, IR, IL)-CRIV*HRIV
          HCOF(IC, IR, IL)=HCOF(IC, IR, IL)-CRIV
          GO TO 100
C
C8-----SINCE HEAD<BOTTOM ADD TERM ONLY TO RHS.
          96 RHS(IC, IR, IL)=RHS(IC, IR, IL)-CRIV*(HRIV-RBOT)
          100 CONTINUE
C
C9-----RETURN
          RETURN
          END

```

List of Variables for Module RIV1FM

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
CRIV	Module	Riverbed conductance.
HCOF	Global	DIMENSION (NCOL,NROW,NLAY), Coefficient of head in the cell (J,I,K) in the finite-difference equation.
HHNEW	Module	HNEW (J,I,K), Single precision.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HRIV	Module	Head in the river.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Column number of the cell containing the river reach.
IL	Module	Layer number of the cell containing the river reach.
IR	Module	Row number of the cell containing the river reach.
L	Module	Index for river reaches.
MXRIVR	Package	Maximum number of river reaches active at any one time.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NRIVER	Package	Number of river reaches active during the current stress period.
NROW	Global	Number of rows in the grid.
RBOT	Module	Temporary field: elevation of the river bottom.
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of the finite-difference equation. RHS is an accumulation of terms from several different packages.
RIVR	Package	DIMENSION (6,MXRIVR), For each reach: layer, row, column, riverhead, riverbed conductance and elevation of bottom of riverbed.

Narrative for Module RIV1BD

This module calculates rates and volumes transferred between the aquifer and rivers.

1. Initialize the cell-by-cell flow-term flag (IBD) and the rate accumulators (RATIN and RATOUT).

2. If there are no reaches ($NRIVER \leq 0$), skip down to step 17 and put zeros into the budget terms for rivers.

3. Test to see if the cell-by-cell flow terms are to be saved on the disk. They will not be saved if either of the following conditions hold: (1) This is not the proper time step ($ICBCFL = 0$) or (2) cell-by-cell flow terms are not to be saved for rivers during this simulation ($IRIVCB \leq 0$). If cell-by-cell flow terms will be saved for this package, set the cell-by-cell flow-term flag (IBD) and clear the buffer in which they will be accumulated (BUFF).

4. For each reach, do steps 5-15 accumulating flows from or into the river.

5. Determine the row, column, and layer of the cell containing the reach.

6. If the cell is external ($IBOUND(I,J,K) \leq 0$), bypass further processing of this reach.

7. Get the river parameters from the river list.

8. Check to see if the head in the cell is greater than the elevation of the bottom of the riverbed.

9. If the head in the cell is greater than the elevation of the bottom of the riverbed, set RATE equal to the conductance of the riverbed times the riverhead minus the head in the cell ($RATE = CRIV*(HRIV - HNEW)$).

10. If the head in the cell is less than or equal to the elevation of the bottom of the riverbed, set RATE equal to the conductance of the riverbed times the riverhead minus the elevation of the bottom of the riverbed ($RATE = CRIV*(HRIV - RBOT)$).

11. If the cell-by-cell flow terms are to be printed, print RATE.

12. If the cell-by-cell flow terms are to be saved, add the RATE to the buffer (BUFF).

13. Check to see whether the flow is into or out of the aquifer.

14. If RATE is negative, add it to RATOUT.

15. If RATE is positive, add it to RATIN.

16. See if the cell-by-cell flow terms are to be saved ($IBD = 1$). If they are, call module UBUDSV to record the buffer (BUFF) onto the disk.

17. Move RATIN and RATOUT into the VBVL array for printing by BAS10T. Add RATIN and RATOUT multiplied by the time-step length to the volume accumulators in VBVL for printing by BAS10T. Move the river budget term labels to VBNM for printing by BAS10T.

18. Increment the budget-term counter (MSUM).

19. RETURN.

Flow Chart for Module RIV1BD

IBD is a flag which, if set, causes cell-by-cell flow terms for river leakage to be recorded.

EXTERNAL: a cell is said to be external if it is either no flow or constant head (i.e., an equation is not formulated for the cell).

RATE is the leakage rate into the aquifer from the river in a cell.

BUFFER is an array in which values are stored as they are being gathered for printing or recording.

RATOUT is an accumulator to which all flows out of the aquifer are added.

RATIN is an accumulator to which all flows into the aquifer are added.

IRIVCB is a flag and a unit number.


If $IRIVCB > 0$, it is the unit number on which cell-by-cell flow terms for rivers will be recorded whenever ICBCFL is set.

If $IRIVCB = 0$, cell-by-cell flow terms for rivers will not be printed or recorded.

If $IRIVCB < 0$, river leakage for each reach will be printed whenever ICBCFL is set.

ICBCFL is a flag.

If $ICBCFL \neq 0$, cell-by-cell flow terms will be printed or recorded (depending on IRIVCB) for the current time step.




```

SUBROUTINE RIV1BD(NRIVER, MXRIVR, RIVR, IBOUND, HNEW,
1      NCOL, NROW, NLAY, DELT, VBVL, VBNM, MSUM, KSTP, KPER, IRIVCB,
2      ICBCFL, BUFF, IOUT)
C-----VERSION 1556 12MAY1987 RIV1BD
C *****
C CALCULATE VOLUMETRIC BUDGET FOR RIVERS
C *****
C
C SPECIFICATIONS:
C -----
C CHARACTER*4 VBNM, TEXT
C DOUBLE PRECISION HNEW
C DIMENSION RIVR(6, MXRIVR), IBOUND(NCOL, NROW, NLAY),
1      HNEW(NCOL, NROW, NLAY), VBVL(4, 20), VBNM(4, 20),
2      BUFF(NCOL, NROW, NLAY)
C DIMENSION TEXT(4)
C DATA TEXT(1), TEXT(2), TEXT(3), TEXT(4) /' R', 'IVER', ' LEA', 'KAGE'/
C -----
C
C1-----INITIALIZE CELL-BY-CELL FLOW TERM FLAG (IBD) AND
C1-----ACCUMULATORS (RATIN AND RATOUT).
      IBD=0
      RATIN=0.
      RATOUT=0.
C
C2-----IF NO REACHES KEEP ZEROES IN ACCUMULATORS.
      IF(NRIVER.EQ.0)GO TO 200
C
C3-----TEST TO SEE IF CELL-BY-CELL FLOW TERMS ARE NEEDED.
      IF(ICBCFL.EQ.0 .OR. IRIVCB.LE.0 ) GO TO 10
C
C3A-----CELL-BY-CELL FLOW TERMS ARE NEEDED SET IBD AND CLEAR BUFFER.
      IBD=1
      DO 5 IL=1, NLAY
      DO 5 IR=1, NROW
      DO 5 IC=1, NCOL
      BUFF(IC, IR, IL)=0.
      5 CONTINUE
C
C4-----FOR EACH RIVER REACH ACCUMULATE RIVER FLOW (STEPS 5-15)
      10 DO 100 L=1, NRIVER
C
C5-----GET LAYER, ROW & COLUMN OF CELL CONTAINING REACH.
      IL=RIVR(1, L)
      IR=RIVR(2, L)
      IC=RIVR(3, L)
C
C6-----IF CELL IS EXTERNAL MOVE ON TO NEXT REACH.
      IF(IBOUND(IC, IR, IL).LE.0)GO TO 100
C
C7-----GET RIVER PARAMETERS FROM RIVER LIST.
      HRIV=RIVR(4, L)

```

```

CRIV=RIVR(5,L)
RBOT=RIVR(6,L)
HHNEW=HNEW(IC,IR,IL)
C
C8-----COMPARE HEAD IN AQUIFER TO BOTTOM OF RIVERBED.
C
C9-----AQUIFER HEAD > BOTTOM THEN RATE=CRIV*(HRIV-HNEW).
      IF(HHNEW.GT.RBOT)RATE=CRIV*(HRIV-HHNEW)
C
C10-----AQUIFER HEAD < BOTTOM THEN RATE=CRIV*(HRIV-RBOT)
      IF(HHNEW.LE.RBOT)RATE=CRIV*(HRIV-RBOT)
C
C11-----PRINT THE INDIVIDUAL RATES IF REQUESTED(IRIVCB<0).
      IF(IRIVCB.LT.0.AND.ICBCFL.NE.0) WRITE(IOUT,900) (TEXT(N),N=1,4),
      1 KPER,KSTP,L,IL,IR,IC,RATE
      900 FORMAT(1H0,4A4,' PERIOD',I3,' STEP',I3,' REACH',I4,
      1 ' LAYER',I3,' ROW',I4,' COL',I4,' RATE',G15.7)
C
C12-----IF C-B-C FLOW TERMS ARE TO BE SAVED THEN ADD RATE TO BUFFER.
      IF(IBD.EQ.1) BUFF(IC,IR,IL)=BUFF(IC,IR,IL)+RATE
C
C13-----SEE IF FLOW IS INTO AQUIFER OR INTO RIVER.
      IF(RATE)94,100,96
C
C14-----AQUIFER IS DISCHARGING TO RIVER SUBTRACT RATE FROM RATOUT.
      94 RATOUT=RATOUT-RATE
      GO TO 100
C
C15-----AQUIFER IS RECHARGED FROM RIVER ADD RATE TO RATIN.
      96 RATIN=RATIN+RATE
      100 CONTINUE
C
C16-----IF C-B-C FLOW TERMS WILL BE SAVED CALL UBUDSV TO RECORD THEM.
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IRIVCB,BUFF,NCOL,NROW,
      1 NLAY,IOUT)
C
C17-----MOVE RATES,VOLUMES & LABELS INTO ARRAYS FOR PRINTING.
      200 VBVL(3,MSUM)=RATIN
      VBVL(4,MSUM)=RATOUT
      VBVL(1,MSUM)=VBVL(1,MSUM)+RATIN*DELT
      VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
C18-----INCREMENT BUDGET TERM COUNTER
      MSUM=MSUM+1
C
C19-----RETURN
      RETURN
      END

```

List of Variables for RIV1BD

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
CRIV	Module	Conductance of the bed of the river reach.
DELT	Global	Length of the current time step.
HHNEW	Module	HNEW (J,I,K), Single precision.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HRIV	Module	Head in the river.
IBD	Module	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be either printed or recorded (depending on IRIVCB) for the current time step.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
IRIVCB	Package	Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, river leakage for each reach will be printed whenever ICBCFL is set.
KPER	Global	Stress period counter.

List of Variables for Module RIV1BD (Continued)

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
KSTP	Global	Time step counter. Reset at the start of each stress period.
L	Module	Index for river reaches.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
MXRIVR	Package	Maximum number of river reaches active at any one time.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NRIVER	Package	Number of river reaches active during the current stress period.
NROW	Global	Number of rows in the grid.
RATE	Module	Flow from the river into the cell. (Reverse the sign to get the flow into the river.)
RATIN	Module	Accumulator for the total flow into the flow field from rivers.
RATOUT	Module	Accumulator for the total flow out of flow field into rivers.
RBOT	Module	Elevation of the bottom of the riverbed.
RIVR	Package	DIMENSION (6,MXRIVR), For each reach: layer, row, column, riverhead, riverbed conductance, and elevation of the bottom of the riverbed.
TEXT	Module	Label to be printed or recorded with the array data.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N), Rate for the current time step into the flow field. (2,N), Rate for the current time step out of the flow field. (3,N), Volume into the flow field during simulation. (4,N), Volume out of the flow field during simulation.

CHAPTER 7

RECHARGE PACKAGE

Conceptualization and Implementation

The Recharge (RCH) Package is designed to simulate areally distributed recharge to the ground-water system. Most commonly, areal recharge occurs as a result of precipitation that percolates to the ground-water system.


Recharge applied to the model is defined as

$$Q_{Ri,j} = I_{i,j} * DELR_j * DELC_i \quad (67)$$

where $Q_{Ri,j}$ is the recharge flow rate applied to the model at horizontal cell location (i,j) expressed as a fluid volume per unit time; and $I_{i,j}$ is the recharge flux (in units of length per unit time) applicable to the map area, $DELR_j * DELC_i$, of the cell. The recharge, $Q_{Ri,j}$, is applied to a single cell within the vertical column of cells located at (i,j) . There is no need to allow for recharge to occur simultaneously at multiple depths in the same vertical column because natural recharge enters the ground-water system at its top. In the simplest situation, the top of the ground-water system will occur in model layer 1; however, the vertical position of the top of the system may vary with horizontal location and with time as the water-table rises and falls. Three options for specifying the cell in each vertical column of cells that receives the recharge have been implemented as described below. The RCH Package can potentially be used to simulate recharge from sources other than precipitation -- for example, artificial recharge. If the ability to apply recharge to more than one cell in a vertical column of cells is required, then the Well Package, which allows recharge or discharge to be specified at any model cell, can be used.


In the package described herein, values of recharge flux, $I_{i,j}$, are read into a two dimensional array, $RECH_{i,j}$, at each stress period (unless an option is exercised to use recharge fluxes from the previous stress period). These values of recharge flux are immediately multiplied by horizontal cell areas, $DEL R_j DEL C_i$, to obtain values of $Q_{R_{i,j}}$, which are maintained in the RECH array. The cell within each vertical column to which the recharge is applied is specified through the recharge option code, NRCHOP, and optional array IRCH. The options include: (1) application of the recharge to model layer 1; (2) application of the recharge to any cell in the vertical column as specified by layer numbers contained in two dimensional array $IRCH_{i,j}$; and (3) application of the recharge to the uppermost active cell in the vertical column, provided there is no constant head cell above it in the column. Under options 1 and 2, if a cell designated to receive recharge is no-flow, then no recharge is added. Under the third option, if there is a constant head cell in a vertical column of cells and there is no active cell above, then no recharge is applied to this column because it is assumed that any recharge would be intercepted by the constant head source. Recharge flux values that are read into the model must be expressed in units that are consistent with the length and time units used to represent all other model parameters.

In the formation of the matrix equations, the recharge flow rate, $Q_{R_{i,j}}$, associated with a given horizontal cell location (i,j) and vertical location, k , that is determined by the recharge option is subtracted from the value of $RHS_{i,j,k}$ (equation (26) or (29)). This is done at each iteration for all cells that receive recharge. Because recharge as defined is independent of aquifer head, nothing is added to the coefficient of head, $HCOF_{i,j,k}$.



Vertical Cross-Section Showing Field Situation With Finite Difference Grid Superimposed


a



Status of Cells at End of Simulation

b


- Variable Head
- Constant Head
- Inactive



Cells Which Receive Recharge Under Option 1

c

- Cell Which Receives Recharge




Cells Which Receive Recharge Under Option 2

d

- Cell Which Receives Recharge
- Inactive Cell Specified by User to Receive Recharge

Heavy Line Encloses Cells User Thought Would Receive Recharge Based on Estimated Water Table



Cells Which Receive Recharge Under Option 3

e

- Cell Which Receives Recharge

Figure 38.—Hypothetical problem showing which cells receive recharge under the three options available in the Recharge Package.

Careful consideration should be given to the problem under study and to the other options employed in its simulation before deciding which of the three recharge options listed above to utilize in a given situation. For example, figure 38 shows a situation in which a cross sectional model has been used to simulate a hypothetical problem involving recharge, seepage from a stream, and seepage into a stream (figure 38-a). Using the provision described in Chapter 5 for horizontal conductance formulation under water table conditions, the model mesh has been progressively truncated during simulation so that the uppermost active cells in each vertical column fall approximately at the water table. This process yields the final distribution of active, constant head and no-flow cells shown in figure 38-b.

Figure 38-c illustrates the recharge distribution to the model if option 1 above is utilized. Under this option recharge is permitted only to the top layer of the model. Thus once, the water table shape has been simulated by the use of no flow cells in the top layer, recharge to the vertical columns beneath those cells is shut off. This clearly fails to simulate the given system.

Figure 38-d illustrates the recharge distribution if option 2 is utilized, assuming that the user specifies recharge cells prior to the simulation on the basis of an estimated water table position, which differs slightly from that finally obtained in the simulation process. Four of the cells which the user had designated as recharge cells have converted to an inactive condition and thus receive no recharge.

Figure 38-e illustrates the Simulation under the third option, which turns out to be the one best suited for this particular situation. Under this option, recharge enters the uppermost active cell in each vertical column, except where constant head cells have been used to represent the streams. Thus, a continuous distribution of recharge to the water table is simulated.

For the typical situation of recharge from precipitation, option 3 is the easiest to use. The model user does not have to be concerned about determining which is the highest active cell in a vertical column because the program automatically determines this throughout the simulation. Option 1, however, can be useful in situations where recharge should not pass through the no-flow cells in layer 1. For example, some cells may be designated no-flow because they are impermeable. Any recharge specified for those cells should not pass into layer 2. Of course option 3 could still be used in this situation by specifying that the recharge rate is zero at the impermeable cells. The user should select the option that will result in the least effort for specification of input data. Similarly, option 2 may be useful when layers other than layer 1 have outcrop areas and when recharge to the specified layers should not penetrate through no-flow cells to a lower layer. Other factors to consider when choosing the best option are that option 2 uses more computer memory than options 1 and 3, and option 3 uses slightly more computer time than options 1 and 2.

Recharge Package Input

Input to the Recharge (RCH) Package is read from the unit specified in IUNIT(8).

FOR EACH SIMULATION

RCH1AL

1. Data: NRCHOP IRCHCB
Format: I10 I10

FOR EACH STRESS PERIOD

RCH1RP

2. Data: INRECH INIRCH
Format: I10 I10
3. Data: RECH(NCOL,NROW)
Module: U2DREL

IF THE RECHARGE OPTION IS EQUAL TO 2

4. Data: IRCH(NCOL,NROW)
Module: U2DINT

Explanation of Fields Used in Input Instructions

NRCHOP--is the recharge option code. Recharge fluxes are defined in a two-dimensional array, RECH, with one value for each vertical column. Accordingly, recharge is applied to one cell in each vertical column, and the option code determines which cell in the column is selected for recharge.

- 1 - Recharge is only to the top grid layer.
- 2 - Vertical distribution of recharge is specified in array IRCH.
- 3 - Recharge is applied to the highest active cell in each vertical column. A constant-head node intercepts recharge and prevents deeper infiltration.

IRCHCB--is a flag and a unit number.

If $IRCHCB > 0$, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set.

If $IRCHCB \leq 0$, cell-by-cell flow terms will not be printed or recorded.

INRECH--is the RECH read flag.

If $INRECH \geq 0$, an array of recharge fluxes, RECH, is read.

If $INRECH < 0$, recharge rates from the preceding stress period are used.

INIRCH--is the IRCH read flag. When NRCHOP is two,

If $INIRCH \geq 0$, an array of layer numbers (IRCH) is read.

If $INIRCH < 0$, the array (IRCH) used in the preceding stress period is reused.

Note: When NRCHOP is one or three, INIRCH is ignored.

RECH--is the recharge flux (Lt^{-1}). Read only if INRECH is greater than or equal to zero.

IRCH--is the layer number array that defines the layer in each vertical column where recharge is applied. Read only if NRCHOP is two and if INIRCH is greater than or equal to zero.

SAMPLE INPUT TO THE RECHARGE PACKAGE USING RECHARGE OPTION 1

DATA ITEM	EXPLANATION	INPUT RECORDS
1	{NRCHOP, IRCHCB}	1 0
2	Stress period 1---{INRECH}	12 3.17E-8 (10F4.0)
3	Control record for recharge array	1.0 1.0 1.0 1.0 1.0
	Recharge rates	1.0 1.0 1.0 1.0 1.1
		1.0 1.0 1.0 1.1 1.1
		1.0 1.0 1.1 1.1 1.1
		1.1 1.1 1.1 1.1 1.1
2	Stress period 2---{INRECH}	-1
2	Stress period 3---{INRECH}	-1
2	Stress period 4---{INRECH}	1
3	Control record for recharge array	12 3.17E-8 (10F4.0)
	Recharge rates	1.2 1.2 1.2 1.2 1.3
		1.2 1.2 1.2 1.3 1.4
		1.2 1.2 1.3 1.4 1.4
		1.0 1.0 1.0 1.1 1.1
		1.2 1.3 1.3 1.4 1.4
		1.3 1.3 1.4 1.4 1.4

SAMPLE INPUT TO THE RECHARGE PACKAGE USING RECHARGE OPTION 2

DATA ITEM	EXPLANATION	INPUT RECORDS
1	{NRCHOP, IRCHCB}	2 0
2	Stress period 1---{INRECH, INIRCH}	1 1
3	Control record for recharge array	0 3.17E-8
4	Control record for layer indicator array	12 1 (20I2)
	Layer numbers	1 2 2 2 3
		1 2 2 2 2
		1 1 2 2 2
		1 1 1 1 2
		1 1 1 1 1
		1 1 1 1 1
2	Stress period 2---{INRECH, INIRCH}	1 1 1 1 1
3	Control record for recharge array	1 -1
2	Stress period 3---{INRECH, INIRCH}	0 1.56E-8
		-1 -1

FIELDS IN ARRAY CONTROL RECORDS ARE---{ LOCAT, CONST, FMTIN, IPRN}

Module Documentation for the Recharge Package

The Recharge Package (RCH1) consists of four modules, all of which are called by the MAIN program. The modules are:

- RCH1AL Allocates space to contain recharge flow rate (RECH) and, if option 2 is specified, the layer-indicator array (IRCH).
- RCH1RP Reads recharge flux (in flow per unit area) and indicator array (if option 2 is specified).
Multiplies recharge flux by cell area.
- RCH1FM Subtract the recharge flow rate from the accumulator in which RHS is formulated.
- RCH1BD Calculates the rate and accumulated volume of recharge into the flow system.

Narrative for Module RCH1AL

This module allocates space in the X array to store data relating to areally distributed recharge.

1. Print a message identifying the package.
2. Read and print the option indicator (NRCHOP) and the unit number for cell-by-cell flow terms (IRCHCB).
3. See if the recharge option (NRCHOP) is legal. If NRCHOP is illegal (not 1, 2, or 3), print a message saying the option is illegal. Do not allocate storage. STOP.
4. If NRCHOP is legal, print NRCHOP.
5. If cell-by-cell flow terms are to be recorded, print the unit number where they will be recorded.
6. Allocate space for the recharge array (RECH). Space is allocated by setting the first element of RECH (LCRECH) equal to the location (ISUM) of the first unused element in the X array and adding the size of the array to ISUM.
7. If the recharge option (NRCHOP) is equal to two, allocate space for a layer-indicator array (IRCH).
8. Calculate and print the number of elements in the X array used by the Recharge Package.
9. RETURN.

Flow Chart for Module RCH1AL


NRCHOP is the recharge option.

- 1 - Recharge is to the top layer.
- 2 - Recharge is to the layer specified by the user in the indicator array (IRCH).
- 3 - Recharge is to the upper-most active cell.

IRCHCB is the unit number on which cell-by-cell flow terms for recharge will be written.

RECH is an array which contains a recharge rate for each horizontal cell location.

IRCH is an array which contains the layer number to which recharge is applied for each horizontal location. It is used only if option 2 has been specified.



```

SUBROUTINE RCH1AL (ISUM, LENX, LCIRCH, LCRECH, NRCHOP,
1          NCOL, NROW, IN, IOUT, IRCHCB)
C
C-----VERSION 1559 12MAY1987 RCH1AL
C *****
C ALLOCATE ARRAY STORAGE FOR RECHARGE
C *****
C
C SPECIFICATIONS:
C -----
C -----
C
C1-----IDENTIFY PACKAGE.
WRITE(IOUT,1)IN
1 FORMAT(1H0,'RCH1 -- RECHARGE PACKAGE, VERSION 1, 9/1/87',
1' INPUT READ FROM UNIT',I3)
C
C2-----READ NRCHOP AND IRCHCB.
READ(IN,2)NRCHOP,IRCHCB
2 FORMAT(2I10)
C
C3-----CHECK TO SEE THAT OPTION IS LEGAL.
IF(NRCHOP.GE.1.AND.NRCHOP.LE.3)GO TO 200
C
C3A-----IF ILLEGAL PRINT A MESSAGE AND ABORT SIMULATION
WRITE(IOUT,8)
8 FORMAT(1X,'ILLEGAL OPTION CODE. SIMULATION ABORTING')
STOP
C
C4-----IF OPTION IS LEGAL PRINT OPTION CODE.
200 IRK=ISUM
IF(NRCHOP.EQ.1) WRITE(IOUT,201)
201 FORMAT(1X,'OPTION 1 -- RECHARGE TO TOP LAYER')
IF(NRCHOP.EQ.2) WRITE(IOUT,202)
202 FORMAT(1X,'OPTION 2 -- RECHARGE TO ONE SPECIFIED NODE IN EACH',
1 ' VERTICAL COLUMN')
IF(NRCHOP.EQ.3) WRITE(IOUT,203)
203 FORMAT(1X,'OPTION 3 -- RECHARGE TO HIGHEST ACTIVE NODE IN EACH',
1 ' VERTICAL COLUMN')
C
C5-----IF CELL-BY-CELL FLOW TERMS TO BE SAVED THEN PRINT UNIT #
IF(IRCHCB.GT.0) WRITE(IOUT,204) IRCHCB
204 FORMAT(1X,'CELL-BY-CELL FLOW TERMS WILL BE RECORDED ON UNIT',I3)
C
C6-----ALLOCATE SPACE FOR THE RECHARGE ARRAY(RECH).
LCRECH=ISUM
ISUM=ISUM+NCOL*NROW
C
C7-----IF OPTION 2 THEN ALLOCATE SPACE FOR INDICATOR ARRAY(IRCH)
LCIRCH=ISUM
IF(NRCHOP.NE.2)GO TO 300
ISUM=ISUM+NCOL*NROW
C
C8-----CALCULATE AND PRINT AMOUNT OF SPACE USED BY RECHARGE.
300 IRK=ISUM-IRK
WRITE(IOUT,4)IRK
4 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED FOR RECHARGE')
ISUM1=ISUM-1
WRITE(IOUT,5)ISUM1,LENX
5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
IF(ISUM1.GT.LENX)WRITE(IOUT,6)
6 FORMAT(1X,' ***X ARRAY MUST BE MADE LARGER***')
C
C9-----RETURN
RETURN
END

```


List of Variables for Module RCH1AL

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
IN	Package	Primary unit number from which input for this package will be read.
IOUT IRCHCB	Global Package	Primary unit number for all printed output. IOUT = 6. Flag. IRCHCB \leq 0, cell-by-cell flow terms will not be printed or recorded. IRCHCB $>$ 0 and ICBCFL \neq 0, cell-by-cell flow terms for the RCH1 Package will be recorded on UNIT = IRCHCB.
IRK	Module	Before this module allocates space, IRK is set equal to ISUM. After allocation, IRK is subtracted from ISUM to get the amount of space in the X array allocated by this module.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1 LCIRCH	Module Package	ISUM-1. Location in the X array of the first element of array IRCH.
LCRECH	Package	Location in the X array of the first element of array RECH.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program.
NCOL NRCHOP	Global Package	Number of columns in the grid. Recharge option: = 1, recharge is to the top grid layer. = 2, recharge is to the grid layer specified in array IRCH. = 3, recharge is to the highest variable-head cell which is not below a constant-head cell.
NROW	Global	Number of rows in the grid.